
はじめに 

 本資料は、RL78/G10を使用する際に、スタートアップルーチンのコードサイ

ズを削減する方法について説明します。 
 

 CA78K0Rのスタートアップルーチンは、RL78で共通の仕様です。 
 

 RL78/G10が、次の仕様であることを考慮した変更方法の説明です。その

ため、他のマイコンを使用した際には正常に動作しない可能性がありますの

で、ご注意ください。 
 RL78-S1コアである 

– 汎用レジスタバンクを搭載していない 
– 外部メモリアクセス機能を搭載していない 

 MAAが0固定 
 RAMサイズが小さい（512バイト） 

 CA78K0Rのスタートアップルーチン内の処理概要 
 RL78/G10向けのスタートアップルーチンのカスタマイズ概要 
 CubeSuite+でのスタートアップルーチンの登録 
 スタートアップルーチンの編集 

CA78K0Rのスタートアップルーチン内の処理概要 

 cstartn.asm の処理の概要は次のようになっています。 

setting the register bank RB 

setting the mirror area 

setting the stack pointer 

call hardware initialization 

clear stack area 

clear saddr area 

copy external variables having initial value (near) 

copy sreg variables which doesn't have initial value 

copy external variables which doesn't have initial value (near) 

copy external variables having initial value (far) 

copy external variables which doesn't have initial value (far) 

copy sreg variables having initial value 

call main function 

RL78/G10向けのスタートアップルーチンのカスタマイズ概要 

 cstartn.asmに対して、次の変更を行います。 

setting the register bank RB 

setting the mirror area 

setting the stack pointer 

call hardware initialization 

clear stack area 

clear saddr area 

copy external variables having initial value (near) 

copy sreg variables which doesn't have initial value 

copy external variables which doesn't have initial value (near) 

copy external variables having initial value (far) 

copy external variables which doesn't have initial value (far) 

copy sreg variables having initial value 

call main function 

clear RAM area 

複数の0クリア処理を、1つの処理にまとめる 

レジスタバンクがないので削除 

0x10000番地以上にミラー領域がないので削除 

外部メモリ機能がないので削除 

外部メモリ機能がないので削除 

コードサイズ 
223バイト → 71バイト 

CubeSuite+でのスタートアップルーチンの登録（1/3） 

 インストールフォルダからスタートアップルーチンのファイルを任意のフォルダ

コピーする 
 コピー対象ファイル 

– cstartn.asm 
– def.inc 
– macro.inc 

 デフォルトのインストールフォルダ（スタートアップ関連ファイル） 
– C:¥Program Files¥Renesas Electronics¥CubeSuite+¥CA78K0R¥バージ

ョン¥Src¥cc78k0r¥src 
（バージョン：CA78K0Rのバージョンです） 

 

6 



© 2014 Renesas Solutions Corp.  

CubeSuite+でのスタートアップルーチンの登録（2/3） 

 コピーしたスタートアップルーチンをCubeSuite+に登録する 

Explorerから、CubeSuite+のプロジェクトツリーパネルの 
「スタートアップ」にcstart.asmをドラッグ＆ドロップする 

CubeSuite+でのスタートアップルーチンの登録（3/3） 

 プロジェクトツリーに登録されたら、編集してください。 

登録完了するとファイルが
表示される 

cstartn.asm が表示されるので、
編集してください 

スタートアップルーチンの編集（1/13） 

; Copyright(c) 2006-2012 Renesas Electronics Corporation 
; RENESAS ELECTRONICS CONFIDENTIAL AND PROPRIETARY. 
; This program must be used solely for the purpose for which 
; it was furnished by Renesas Electronics Corporation. No part of this 
; program may be reproduced or disclosed to others, in any 
; form, without the prior written permission of Renesas Electronics 
; Corporation. 
;------------------------------------- 
; W-1 cstart (no init) 
; 
;------------------------------------- 
 NAME @cstart 
 
$INCLUDE (def.inc) 
$INCLUDE (macro.inc) 
 

変更なし 

スタートアップルーチンの編集（2/13） 

 

10 

;----------------------------------------------------------------------------- 
; declaration of symbol 
; 
; attention): change to EQU value 1 -> 0  if necessary 
;----------------------------------------------------------------------------- 
; 
FLOATSW  EQU 1 ;floating point variables use 
 
 PUBLIC _@cstart,_@cend 
 
$_IF(FLOATSW) 
 PUBLIC _errno 
$ENDIF 
 

変更なし 

スタートアップルーチンの編集（3/13） 

 

11 

;----------------------------------------------------------------------------- 
; external declaration of symbol for stack area 
; 
; _@STBEG has value of the end address +1 of compiler's stack area. 
; _@STEND has value of the start address of compiler's stack area. 
; _@STBEG and _@STEND are created by linker with -S option. 
; Accordingly, specify the -S option when linking. 
; 
; _@MAA has value of the mirror area. 
; _@SADSIZ has value of the size of the saddr area. 
; _@SADBEG has value of the start address of the saddr area. 
;----------------------------------------------------------------------------- 
 EXTRN _main,_@STBEG,_@STEND,_stkinit,_hdwinit,_@SADSIZ,_@SADBEG 
 
$_IF(_MIRROR) 
 EXTRN _@MAA 
$ENDIF 
 
 EXTRN _?R_INIT,_?RLINIT,_?R_INIS,_?DATA,_?DATAL,_?DATS 
 

削除 

スタートアップルーチンの編集（4/13） 

;----------------------------------------------------------------------------- 
; allocation area which library uses 
; _errno errno number code 
;----------------------------------------------------------------------------- 
@@DATA DSEG BASEP ;near 
_@DATA: 
$_IF(FLOATSW) 
_errno:  DS 2 
$ENDIF 
 
@@VECT00 CSEG AT 0 
 DW _@cstart 
 
@@LCODE CSEG BASE 
 
_@cstart: 
 

変更なし 

スタートアップルーチンの編集（5/13） 

;----------------------------------------------------------------------------- 
; setting the register bank RB0 as work register set 
;----------------------------------------------------------------------------- 
 SEL RB0 
;----------------------------------------------------------------------------- 
; setting the mirror area 
;----------------------------------------------------------------------------- 
$_IF(_MIRROR) 
 MOV A,#_@MAA 
 MOV1 CY,A.0 
 MOV1 MAA,CY 
$ENDIF 
 
;----------------------------------------------------------------------------- 
; setting the stack pointer 
; 
; _@STBEG is created by linker with -S option. 
;----------------------------------------------------------------------------- 
 MOVW SP,#LOWW _@STBEG ;SP <- stack begin address 
 
;----------------------------------------------------------------------------- 
; hardware initialization 
;----------------------------------------------------------------------------- 
 CALL !!_hdwinit 
 
 

削除 

削除 

スタートアップルーチンの編集（6/13） 

;----------------------------------------------------------------------------- 
; clear stack area 
; 
; _@STEND is created by linker with -S option. 
;----------------------------------------------------------------------------- 
 MOVW HL,#LOWW _@STEND ;HL <- stack end address 
 CALL !!_stkinit 
 
;----------------------------------------------------------------------------- 
; clear saddr 
;----------------------------------------------------------------------------- 
 MOV B,#_@SADSIZ ;0FEDFH - _@SADBEG + 1 
 CLRW AX 
LSADR1: 
 DEC B 
 DEC B 
 MOVW _@SADBEG[B],AX 
 BNZ $LSADR1 
 

削除 

削除 

スタートアップルーチンの編集（7/13） 

;----------------------------------------------------------------------------- 
; clear Internal RAM ; add for RL78/G10 
;----------------------------------------------------------------------------- 
 MOVW HL,#LOWW 0FFDE0H ; Start address of Internal RAM 
 MOVW AX,#LOWW 0FFEE0H ; End address of Internal RAM + 1 
 BR $LDATA2 
LDATA1: 
 MOV [HL+0],#0 
 INCW HL 
LDATA2: 
 CMPW AX,HL 
 BNZ $LDATA1 

追加 



© 2014 Renesas Solutions Corp.  

スタートアップルーチンの編集（8/13） 

;----------------------------------------------------------------------------- 
; ROM data copy 
;----------------------------------------------------------------------------- 
 
; copy external variables having initial value 
$_IF(_ESCOPY) 
 MOV ES,#HIGHW _@R_INIT 
$ENDIF 
 MOVW HL,#LOWW _@R_INIT 
 MOVW DE,#LOWW _@INIT 
 BR $LINIT2 
LINIT1: 
$_IF(_ESCOPY) 
 MOV A,ES:[HL] 
$ELSE 
 MOV A,[HL] 
$ENDIF 
 MOV [DE],A 
 INCW HL 
 INCW DE 
LINIT2: 
 MOVW AX,HL 
 CMPW AX,#LOWW _?R_INIT 
 BNZ $LINIT1 

変更なし 

スタートアップルーチンの編集（9/13） 

$_IF(_DATAFAR) 
; copy external variables having initial value 
 MOVW HL,#LOWW _@RLINIT 
 MOVW DE,#LOWW _@INITL 
 BR $LINIT4 
LINIT3: 
 MOV ES,#HIGHW _@RLINIT 
 MOV A,ES:[HL] 
 MOV ES,#HIGHW _@INITL 
 MOV ES:[DE],A 
 INCW HL 
 INCW DE 
LINIT4: 
 MOVW AX,HL 
 CMPW AX,#LOWW _?RLINIT 
 BNZ $LINIT3 
$ENDIF 

削除 

スタートアップルーチンの編集（10/13） 

; copy external variables which doesn't have initial value 
 MOVW HL,#LOWW _@DATA 
 MOVW AX,#LOWW _?DATA 
 BR $LDATA2 
LDATA1: 
 MOV [HL+0],#0 
 INCW HL 
LDATA2: 
 CMPW AX,HL 
 BNZ $LDATA1 
$_IF(_DATAFAR) 
; copy external variables which doesn't have initial value 
 MOV ES,#HIGHW _@DATAL 
 MOVW HL,#LOWW _@DATAL 
 MOVW AX,#LOWW _?DATAL 
 BR $LDATA4 
LDATA3: 
 MOV ES:[HL+0],#0 
 INCW HL 
LDATA4: 
 CMPW AX,HL 
 BNZ $LDATA3 
$ENDIF 

削除 

削除 

スタートアップルーチンの編集（11/13） 

; copy sreg variables having initial value 
$_IF(_ESCOPY) 
 MOV ES,#HIGHW _@R_INIS 
$ENDIF 
 MOVW HL,#LOWW _@R_INIS 
 MOVW DE,#LOWW _@INIS 
 BR $LINIS2 
LINIS1: 
$_IF(_ESCOPY) 
 MOV A,ES:[HL] 
$ELSE 
 MOV A,[HL] 
$ENDIF 
 MOV [DE],A 
 INCW HL 
 INCW DE 
LINIS2: 
 MOVW AX,HL 
 CMPW AX,#LOWW _?R_INIS 
 BNZ $LINIS1 
 

変更なし 

スタートアップルーチンの編集（12/13） 

; copy sreg variables which doesn't have initial value 
 MOVW HL,#LOWW _@DATS 
 MOVW AX,#LOWW _?DATS 
 BR $LDATS2 
LDATS1: 
 MOV [HL+0],#0 
 INCW HL 
LDATS2: 
 CMPW AX,HL 
 BNZ $LDATS1 
 
;----------------------------------------------------------------------------- 
; call main function 
;----------------------------------------------------------------------------- 
 CALL !!_main  ;main(); 
 BR $$ 
; 
_@cend: 
 

削除 

スタートアップルーチンの編集（13/13） 

;----------------------------------------------------------------------------- 
; define segment and label used by ROMable processing 
;----------------------------------------------------------------------------- 
@@R_INIT CSEG UNIT64KP 
_@R_INIT: 
$_IF(_DATAFAR) 
@@RLINIT CSEG UNIT64KP 
_@RLINIT: 
$ENDIF 
@@R_INIS CSEG UNIT64KP 
_@R_INIS: 
@@INIT DSEG BASEP 
_@INIT: 
$_IF(_DATAFAR) 
@@INITL DSEG UNIT64KP 
_@INITL: 
$ENDIF 
$_IF(_DATAFAR) 
@@DATAL DSEG UNIT64KP 
_@DATAL: 
$ENDIF 
@@INIS DSEG SADDRP 
_@INIS: 
@@DATS DSEG SADDRP 
_@DATS: 
@@CALT CSEG CALLT0 
@@CNST CSEG MIRRORP 
@@CNSTL CSEG PAGE64KP 
@@BITS BSEG 
; 
 END 

変更なし 

